Rules for integrands of the form $(d + e\,x)^m\,(a + b\,\text{ArcTanh}[c\,x^n])^p$

1. $\int (d + e\,x)^q\,(a + b\,\text{ArcTanh}[c\,x])^p\,dx$ when $p \in \mathbb{Z}^+$

   1. $\int \frac{(a + b\,\text{ArcTanh}[c\,x])^p}{d + e\,x}\,dx$ when $p \in \mathbb{Z}^+$

      1: $\int \frac{(a + b\,\text{ArcTanh}[c\,x])^p}{d + e\,x}\,dx$ when $p \in \mathbb{Z}^+ \wedge c^2\,d^2 - e^2 == 0$

Derivation: Integration by parts

Basis: $\frac{1}{d+e\,x} == -\frac{1}{e}\,\partial_x \text{Log}\left[\frac{2}{1+\frac{e\,x}{d}}\right]$

Rule: If $p \in \mathbb{Z}^+ \wedge c^2\,d^2 - e^2 == 0$, then

$$\int \frac{(a + b\,\text{ArcTanh}[c\,x])^p}{d + e\,x}\,dx \longrightarrow -\frac{(a + b\,\text{ArcTanh}[c\,x])^p\,\text{Log}\left[\frac{2}{1+\frac{e\,x}{d}}\right]}{e} + \frac{b\,c\,p}{e}\int \frac{(a + b\,\text{ArcTanh}[c\,x])^{p-1}\,\text{Log}\left[\frac{2}{1+\frac{e\,x}{d}}\right]}{1 - c^2\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTanh[c*x])^p*Log[2/(1+e*x/d)]/e +
  b*c*p/e*Int[(a+b*ArcTanh[c*x])^(p-1)*Log[2/(1+e*x/d)]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2-e^2,0]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCoth[c*x])^p*Log[2/(1+e*x/d)]/e +
  b*c*p/e*Int[(a+b*ArcCoth[c*x])^(p-1)*Log[2/(1+e*x/d)]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2-e^2,0]
```

2. $\int \frac{(a+b\,\text{ArcTanh}[c\,x])^p}{d+e\,x}\,dx$ when $p\in\mathbb{Z}^+\land c^2 d^2-e^2\neq 0$

1: $\int \frac{a+b\,\text{ArcTanh}[c\,x]}{d+e\,x}\,dx$ when $c^2 d^2-e^2\neq 0$

Derivation: Algebraic expansion and integration by parts

Basis: $\frac{1}{d+e\,x} == \frac{c}{e\,(1+c\,x)} - \frac{c\,d-e}{e\,(1+c\,x)\,(d+e\,x)}$

Basis: $\frac{1}{1+c\,x} == -\frac{1}{c}\,\partial_x \text{Log}\left[\frac{2}{1+c\,x}\right]$

Basis: $\frac{1}{(1+c\,x)\,(d+e\,x)} == -\frac{1}{c\,d-e}\,\partial_x \text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]$

Basis: $\partial_x\,(a+b\,\text{ArcTanh}[c\,x]) == \frac{b\,c}{1-c^2 x^2}$

Rule: If $c^2 d^2-e^2\neq 0$, then

$$\int \frac{a+b\,\text{ArcTanh}[c\,x]}{d+e\,x}\,dx \longrightarrow \frac{c}{e}\int \frac{a+b\,\text{ArcTanh}[c\,x]}{1+c\,x}\,dx - \frac{c\,d-e}{e}\int \frac{a+b\,\text{ArcTanh}[c\,x]}{(1+c\,x)\,(d+e\,x)}\,dx \longrightarrow$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{b\,c}{e}\int \frac{\text{Log}\left[\frac{2}{1+c\,x}\right]}{1-c^2 x^2}\,dx + \frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e} - \frac{b\,c}{e}\int \frac{\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{1-c^2 x^2}\,dx \longrightarrow$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{b\,\text{PolyLog}\left[2,1-\frac{2}{1+c\,x}\right]}{2\,e} + \frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e} - \frac{b\,\text{PolyLog}\left[2,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{2\,e}$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTanh[c*x])*Log[2/(1+c*x)]/e +
  b*c/e*Int[Log[2/(1+c*x)]/(1-c^2*x^2),x] +
  (a+b*ArcTanh[c*x])*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b*c/e*Int[Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCoth[c*x])*Log[2/(1+c*x)]/e +
  b*c/e*Int[Log[2/(1+c*x)]/(1-c^2*x^2),x] +
  (a+b*ArcCoth[c*x])*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b*c/e*Int[Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

2: $\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2}{d+e\,x}\,dx$ when $c^2 d^2 - e^2 \neq 0$

Derivation: Algebraic expansion and integration by parts

Basis: $\frac{1}{d+e\,x} = \frac{c}{e\,(1+c\,x)} - \frac{c\,d-e}{e\,(1+c\,x)\,(d+e\,x)}$

Basis: $\frac{1}{1+c\,x} = -\frac{1}{c}\,\partial_x \text{Log}\left[\frac{2}{1+c\,x}\right]$

Basis: $\frac{1}{(1+c\,x)\,(d+e\,x)} = -\frac{1}{c\,d-e}\,\partial_x \text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]$

Basis: $\partial_x\,(a+b\,\text{ArcTanh}[c\,x])^2 = \frac{2\,b\,c\,(a+b\,\text{ArcTanh}[c\,x])}{1-c^2\,x^2}$

Rule: If $c^2 d^2 - e^2 \neq 0$, then

$$\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2}{d+e\,x}\,dx \longrightarrow \frac{c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2}{1+c\,x}\,dx - \frac{c\,d-e}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2}{(1+c\,x)\,(d+e\,x)}\,dx \longrightarrow$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{2\,b\,c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{1-c^2\,x^2}\,dx +$$

$$\frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e} - \frac{2\,b\,c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{1-c^2\,x^2}\,dx \longrightarrow$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{b\,(a+b\,\text{ArcTanh}[c\,x])\,\text{PolyLog}\left[2,\,1-\frac{2}{1+c\,x}\right]}{e} + \frac{b^2\,\text{PolyLog}\left[3,\,1-\frac{2}{1+c\,x}\right]}{2\,e} +$$

$$\frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e}-\frac{b\,(a+b\,\text{ArcTanh}[c\,x])\,\text{PolyLog}\left[2,\,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e}-\frac{b^2\,\text{PolyLog}\left[3,\,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{2\,e}$$

–

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])^2/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTanh[c*x])^2*Log[2/(1+c*x)]/e +
  b*(a+b*ArcTanh[c*x])*PolyLog[2,1-2/(1+c*x)]/e +
  b^2*PolyLog[3,1-2/(1+c*x)]/(2*e) +
   (a+b*ArcTanh[c*x])^2*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b*(a+b*ArcTanh[c*x])*PolyLog[2,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b^2*PolyLog[3,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])^2/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCoth[c*x])^2*Log[2/(1+c*x)]/e +
  b*(a+b*ArcCoth[c*x])*PolyLog[2,1-2/(1+c*x)]/e +
  b^2*PolyLog[3,1-2/(1+c*x)]/(2*e) +
   (a+b*ArcCoth[c*x])^2*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b*(a+b*ArcCoth[c*x])*PolyLog[2,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  b^2*PolyLog[3,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

3: $\int \frac{(a+b\,\text{ArcTanh}[c\,x])^3}{d+e\,x}\,dx$ when $c^2 d^2 - e^2 \neq 0$

Derivation: Algebraic expansion and integration by parts

Basis: $\frac{1}{d+e\,x} == \frac{c}{e\,(1+c\,x)} - \frac{c\,d-e}{e\,(1+c\,x)\,(d+e\,x)}$

Basis: $\frac{1}{1+c\,x} == -\frac{1}{c}\,\partial_x \text{Log}\left[\frac{2}{1+c\,x}\right]$

Basis: $\frac{1}{(1+c\,x)\,(d+e\,x)} == -\frac{1}{c\,d-e}\,\partial_x \text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]$

Basis: $\partial_x\,(a+b\,\text{ArcTanh}[c\,x])^3 == \frac{3\,b\,c\,(a+b\,\text{ArcTanh}[c\,x])^2}{1-c^2\,x^2}$

– Rule: If $c^2 d^2 - e^2 \neq 0$, then

$$\int \frac{(a+b\,\text{ArcTanh}[c\,x])^3}{d+e\,x}\,dx \to \frac{c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^3}{1+c\,x}\,dx - \frac{c\,d-e}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^3}{(1+c\,x)\,(d+e\,x)}\,dx \to$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])^3\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{3\,b\,c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{1-c^2\,x^2}\,dx +$$
$$\frac{(a+b\,\text{ArcTanh}[c\,x])^3\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e} - \frac{3\,b\,c}{e}\int \frac{(a+b\,\text{ArcTanh}[c\,x])^2\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{1-c^2\,x^2}\,dx \to$$

$$-\frac{(a+b\,\text{ArcTanh}[c\,x])^3\,\text{Log}\left[\frac{2}{1+c\,x}\right]}{e} + \frac{3\,b\,(a+b\,\text{ArcTanh}[c\,x])^2\,\text{PolyLog}\left[2,1-\frac{2}{1+c\,x}\right]}{2\,e} +$$
$$\frac{3\,b^2\,(a+b\,\text{ArcTanh}[c\,x])\,\text{PolyLog}\left[3,1-\frac{2}{1+c\,x}\right]}{2\,e} + \frac{3\,b^3\,\text{PolyLog}\left[4,1-\frac{2}{1+c\,x}\right]}{4\,e} +$$
$$\frac{(a+b\,\text{ArcTanh}[c\,x])^3\,\text{Log}\left[\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{e} - \frac{3\,b\,(a+b\,\text{ArcTanh}[c\,x])^2\,\text{PolyLog}\left[2,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{2\,e} -$$
$$\frac{3\,b^2\,(a+b\,\text{ArcTanh}[c\,x])\,\text{PolyLog}\left[3,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{2\,e} - \frac{3\,b^3\,\text{PolyLog}\left[4,1-\frac{2\,c\,(d+e\,x)}{(c\,d+e)\,(1+c\,x)}\right]}{4\,e}$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])^3/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcTanh[c*x])^3*Log[2/(1+c*x)]/e +
  3*b*(a+b*ArcTanh[c*x])^2*PolyLog[2,1-2/(1+c*x)]/(2*e) +
  3*b^2*(a+b*ArcTanh[c*x])*PolyLog[3,1-2/(1+c*x)]/(2*e) +
  3*b^3*PolyLog[4,1-2/(1+c*x)]/(4*e) +
  (a+b*ArcTanh[c*x])^3*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  3*b*(a+b*ArcTanh[c*x])^2*PolyLog[2,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) -
  3*b^2*(a+b*ArcTanh[c*x])*PolyLog[3,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) -
  3*b^3*PolyLog[4,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(4*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])^3/(d_+e_.*x_),x_Symbol] :=
  -(a+b*ArcCoth[c*x])^3*Log[2/(1+c*x)]/e +
  3*b*(a+b*ArcCoth[c*x])^2*PolyLog[2,1-2/(1+c*x)]/(2*e) +
  3*b^2*(a+b*ArcCoth[c*x])*PolyLog[3,1-2/(1+c*x)]/(2*e) +
  3*b^3*PolyLog[4,1-2/(1+c*x)]/(4*e) +
  (a+b*ArcCoth[c*x])^3*Log[2*c*(d+e*x)/((c*d+e)*(1+c*x))]/e -
  3*b*(a+b*ArcCoth[c*x])^2*PolyLog[2,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) -
  3*b^2*(a+b*ArcCoth[c*x])*PolyLog[3,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(2*e) -
  3*b^3*PolyLog[4,1-2*c*(d+e*x)/((c*d+e)*(1+c*x))]/(4*e) /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d^2-e^2,0]
```

2: $\int (d+e\,x)^q\,(a+b\,\text{ArcTanh}[c\,x])\,dx$ when $q\neq -1$

Derivation: Integration by parts

Rule: If $q\neq -1$, then

$$\int (d+e\,x)^q\,(a+b\,\text{ArcTanh}[c\,x])\,dx \longrightarrow \frac{(d+e\,x)^{q+1}\,(a+b\,\text{ArcTanh}[c\,x])}{e\,(q+1)}-\frac{b\,c}{e\,(q+1)}\int\frac{(d+e\,x)^{q+1}}{1-c^2\,x^2}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcTanh[c_.*x_]),x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcTanh[c*x])/(e*(q+1)) -
  b*c/(e*(q+1))*Int[(d+e*x)^(q+1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcCoth[c_.*x_]),x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcCoth[c*x])/(e*(q+1)) -
  b*c/(e*(q+1))*Int[(d+e*x)^(q+1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

3: $\int (d+e\,x)^q\,(a+b\,\text{ArcTanh}[c\,x])^p\,dx$ when $p-1\in\mathbb{Z}^+\land q\in\mathbb{Z}\land q\neq -1$

Derivation: Integration by parts

Rule: If $p-1\in\mathbb{Z}^+\land q\in\mathbb{Z}\land q\neq -1$, then

$$\int (d+e\,x)^q\,(a+b\,\text{ArcTanh}[c\,x])^p\,dx\,\longrightarrow$$
$$\frac{(d+e\,x)^{q+1}\,(a+b\,\text{ArcTanh}[c\,x])^p}{e\,(q+1)}-\frac{b\,c\,p}{e\,(q+1)}\int (a+b\,\text{ArcTanh}[c\,x])^{p-1}\,\text{ExpandIntegrand}\left[\frac{(d+e\,x)^{q+1}}{1-c^2\,x^2},\,x\right]dx$$

Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcTanh[c_.*x_])^p_,x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcTanh[c*x])^p/(e*(q+1)) -
  b*c*p/(e*(q+1))*Int[ExpandIntegrand[(a+b*ArcTanh[c*x])^(p-1),(d+e*x)^(q+1)/(1-c^2*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,1] && IntegerQ[q] && NeQ[q,-1]
```

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*ArcCoth[c_.*x_])^p_,x_Symbol] :=
  (d+e*x)^(q+1)*(a+b*ArcCoth[c*x])^p/(e*(q+1)) -
  b*c*p/(e*(q+1))*Int[ExpandIntegrand[(a+b*ArcCoth[c*x])^(p-1),(d+e*x)^(q+1)/(1-c^2*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,1] && IntegerQ[q] && NeQ[q,-1]
```

2. $\int (d+e\,x)^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p dx$ when $p \in \mathbb{Z}^+$

1. $\int (d+e\,x)^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right) dx$

1. $\int \frac{a+b\,\text{ArcTanh}\left[c\,x^n\right]}{d+e\,x} dx$

1: $\int \frac{a+b\,\text{ArcTanh}\left[c\,x^n\right]}{d+e\,x} dx$ when $n \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right) == b\,c\,n\,\frac{x^{n-1}}{1-c^2\,x^{2\,n}}$

Rule: If $n \in \mathbb{Z}$, then

$$\int \frac{a+b\,\text{ArcTanh}\left[c\,x^n\right]}{d+e\,x} dx \longrightarrow \frac{\text{Log}[d+e\,x]\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)}{e} - \frac{b\,c\,n}{e} \int \frac{x^{n-1}\,\text{Log}[d+e\,x]}{1-c^2\,x^{2\,n}} dx$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_^n_])/(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*ArcTanh[c*x^n])/e -
  b*c*n/e*Int[x^(n-1)*Log[d+e*x]/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[n]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_^n_])/(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*ArcCoth[c*x^n])/e -
  b*c*n/e*Int[x^(n-1)*Log[d+e*x]/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[n]
```

2: $\int \frac{a + b\,\text{ArcTanh}[c\,x^n]}{d + e\,x}\,dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] == k\,\text{Subst}\left[x^{k-1}\,F\left[x^k\right],\,x,\,x^{1/k}\right]\,\partial_x x^{1/k}$

Rule: If $n \in \mathbb{F}$, let $k \to \text{Denominator}[n]$, then

$$\int \frac{a + b\,\text{ArcTanh}[c\,x^n]}{d + e\,x}\,dx \longrightarrow k\,\text{Subst}\left[\int \frac{x^{k-1}\left(a + b\,\text{ArcTanh}\left[c\,x^{k\,n}\right]\right)}{d + e\,x^k}\,dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcTanh[c*x^(k*n)])/(d+e*x^k),x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_^n_])/(d_+e_.*x_),x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcCoth[c*x^(k*n)])/(d+e*x^k),x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n]
```

2: $\int (d + e\,x)^m \left(a + b\,\text{ArcTanh}[c\,x^n]\right)\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \left(a + b\,\text{ArcTanh}[c\,x^n]\right) == b\,c\,n\,\frac{x^{n-1}}{1-c^2\,x^{2\,n}}$

Rule: If $m \neq -1$, then

$$\int (d+e\,x)^m\,(a+b\,\text{ArcTanh}[c\,x^n])\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcTanh}[c\,x^n])}{e\,(m+1)}-\frac{b\,c\,n}{e\,(m+1)}\int\frac{x^{n-1}\,(d+e\,x)^{m+1}}{1-c^2\,x^{2\,n}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcTanh[c*x^n])/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[x^(n-1)*(d+e*x)^(m+1)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCoth[c*x^n])/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[x^(n-1)*(d+e*x)^(m+1)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{ArcTanh}[c\,x^n])^p\,dx$ when $p-1\in\mathbb{Z}^+\land m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p-1\in\mathbb{Z}^+\land m\in\mathbb{Z}^+$, then

$$\int (d+e\,x)^m\,(a+b\,\text{ArcTanh}[c\,x^n])^p\,dx \longrightarrow \int (a+b\,\text{ArcTanh}[c\,x^n])^p\,\text{ExpandIntegrand}[(d+e\,x)^m,\,x]\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTanh[c*x^n])^p,(d+e*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,1] && IGtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCoth[c*x^n])^p,(d+e*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,1] && IGtQ[m,0]
```

U: $\int (d+e\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx$

Rule:

$$\int (d+e\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx \longrightarrow \int (d+e\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(a+b*ArcTanh[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*(a+b*ArcCoth[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```